COMITATO NAZIONALE PER L'ENERGIA NUCLEARE
Laboratori Nazionali di Frascati

LNF-62/22

M.G. Trigila-Cao: SOLUZIONE NUMERICA DELLE EQUAZIONI INTEGRALI DI VOLTERRA MEDIANTE CALCOLATORE I.B.M. 1620.

Nota interna: n° 125
5 Aprile 1962

Servizio Documentazione
dei Laboratori Nazionali di Frascati del CNEN
Casella Postale 70 - Frascati (Roma)


START
Leggi: H, x0, N
AαB, θβ
Y0(x0) = f(x0)
p = 0
p1 = 0
xp = x0 + pH
α = 1
x = xp + θα H
Fp(x) = f(x)
β = 0
p ≥ 1
sì
no
Y0(xp) = Y4(xp-1)
j = 0
j1 = 0
xj = x0 + jH
Fp(x) = Fp(x) + HBβ G[x, xj + θβ H, YB(xj)]
β = β + 1
β ≤ 3
β = 0
j = j + 1
j1 = j1 + 1
j ≤ p-1
S = 0
S = S + AαβG[x, xp + θβ H, YB(xj)]
β = β + 1
β > α - 1
Yα(xp) = HS + Fp(x)
α = α + 1
α > 4
stampa:
x, Y4(xp)
p = p + 1
p > N
p1 = p1 + 1

Laboratori Nazionali di Frascati del C.N.E.N.
Servizio Documentazione

LNF-62/22

Nota interna: n° 125
5 Aprile 1962

M.G. Trigila-Cao: SOLUZIONE NUMERICA DELLE EQUAZIONI INTEGRALI DI VOLTERRA MEDIANTE CALCOLATORE I.B.M. 1620.

Fondamenti matematici

Si consideri l'equazione di Volterra:

$$\varphi(x) = f(x) + \int_{x_0}^{x} G[x, s, \varphi(s)]\, ds \tag{0}$$

dove la f e la G sono funzioni regolari date.

La sua soluzione numerica può essere affrontata con un metodo analogo a quello di Runge e Kutta per le equazioni differenziali ordinarie.

Infatti se si pone:

$$\varphi_\alpha(x_p) = F_p(x_p + \theta_\alpha h) + h \sum_{\beta=0}^{\alpha-1} A_{\alpha\beta}\, G[x_p + \theta_\alpha h, x_p + \theta_\beta h, \varphi_\beta(x_p)] \tag{1}$$

$$\sum_{\beta=0}^{\alpha-1} A_{\alpha\beta} = \theta_\alpha \tag{2}$$

dove: p = 0, 1, 2,......n- 1

n = numero dei punti in cui si vuol calcolare $\varphi(x)$

$\alpha = 1, 2, 3, 4$

$x_p = x_0 + p\,h$

h = passo usato durante l'elaborazione

(3) $\qquad F_0(x) = f(x)$

e per $p \geq 1$

(4) $\qquad \varphi_0(x_p) = \varphi_4(x_{p-1})$

(5) $$F_p(x) = f(x) + h \sum_{j=0}^{p-1} \sum_{\beta=0}^{3} A_{4\beta}\, G[x, x_j + \theta_\beta h, \varphi_\beta(x_j)]$$

e si fa in modo che lo sviluppo in serie di Taylor di $\varphi(x_0 + h) - \varphi_4(x_0)$ inizi con il termine in h^5 si ottengono per i coefficienti $A_{\alpha\beta}$ e θ_j ($j = 0, 1, 2, 3, 4$) le stesse relazioni che si trovano per le equazioni differenziali ordinarie. Occorre precisare che tali relazioni non determinano univocamente i suddetti coefficienti; imponendo, allora, condizioni analoghe a quelle che si impongono per la deduzione dei coefficienti di Runge, Kutta o Gill si ottengono per essi tre possibili gruppi di valori numerici.

Operando alla Runge si ha:

$\theta_0 = 0$; $\theta_1 = \frac{1}{2}$; $\theta_2 = \frac{1}{2}$; $\theta_3 = 1$; $\theta_4 = 1$; $A_{10} = \frac{1}{2}$; $A_{20} = 0$; $A_{21} = \frac{1}{2}$;

$A_{30} = 0$; $A_{31} = 0$; $A_{32} = 1$; $A_{40} = \frac{1}{6}$; $A_{41} = \frac{1}{3}$; $A_{42} = \frac{1}{3}$; $A_{43} = \frac{1}{6}$.

Operando alla Kutta si ottiene:

$\theta_0 = 0$; $\theta_1 = \frac{1}{3}$; $\theta_2 = \frac{2}{3}$; $\theta_3 = 1$; $\theta_4 = 1$; $A_{10} = \frac{1}{3}$; $A_{20} = -\frac{1}{3}$;

$A_{21} = 1$; $A_{30} = 1$; $A_{31} = -1$; $A_{32} = 1$; $A_{40} = \frac{1}{8}$; $A_{41} = \frac{3}{8}$;

$A_{42} = \frac{3}{8}$; $A_{43} = \frac{1}{8}$.

Operando alla Gill si ha:

$$\vartheta_0 = 0;\ \vartheta_1 = \frac{1}{2};\ \vartheta_2 = \frac{1}{2};\ \vartheta_3 = 1;\ \theta_4 = 1;\ A_{10} = \frac{1}{2};\ A_{20} = -\frac{1}{2} + \frac{1}{\sqrt{2}};$$

$$A_{21} = 1 - \frac{1}{\sqrt{2}};\ A_{30} = 0;\ A_{31} = -\frac{1}{\sqrt{2}};\ A_{32} = 1 + \frac{1}{\sqrt{2}};\ A_{40} = \frac{1}{6};$$

$$A_{41} = \frac{1}{3}\left(1 - \frac{1}{\sqrt{2}}\right);\ A_{42} = \frac{1}{3}\left(1 + \frac{1}{\sqrt{2}}\right);\ A_{43} = \frac{1}{6}.$$

L'errore che si commette risulta sempre minore o uguale ad h^5.

Le formule che servono per la soluzione numerica della (0) sono: la (1), la (3), la (4) e la (5).

Descrizione del programma

Il programma è stato eseguito in S.P.S. seguendo lo schema indicato nell'annesso diagramma a blocchi.

Esso è stato fatto in modo che la sua applicazione a qualunque equazione integrale del tipo (0) non presenti nessuna difficoltà: basterà, infatti, cambiare, nel modo che ora specificheremo le istruzioni che si riferiscono al calcolo delle funzioni f(x) e $G[x, s, \varphi(s)]$ (il programma allegato si riferisce all'equaz. b (cfr. esempi numerici)) e modificare l'istruzione di riserva delle memorie.

Le istruzioni che vanno dal label CALF al label CALG (escluso) si riferiscono al calcolo della funzione f(x). L'argomento x si trova all'indirizzo xA e il valore della funzione deve essere posto all'indirizzo L.

Le istruzioni che vanno dal label CALG al label RIPRI (escluso) si riferiscono al calcolo della funzione $G[x, s, \varphi(s)]$.

In xB si trova il valore attuale della variabile s, mentre in BG si trova l'indirizzo della φ(s) corrispondente. Infine in G va posto il valore della funzione $G[x, s, \varphi(s)]$.

H è il passo che viene usato durante l'elaborazione, x0 è l'estremo inferiore dell'integrale che figura nell'equazione di Volterra, N è il numero di punti (diminuito di uno) in cui si vuol calcolare la $\varphi(x)$.

L'istruzione che riserva il numero delle memorie occorrenti per eseguire il calcolo (dipendente dal numero di punti in cui si vuol calcolare la $\varphi(x)$) è il DSB avente per label FI. Nel nostro caso si è posto N = 19, quindi tenendo conto anche dei valori intermedi in cui viene calcolata la $\varphi(x)$, il numero di elementi del blocco FI risulta (N+2) x 4+1 = 85. Per esempio, se si volesse calcolare la $\varphi(x)$ in 50 punti si avrebbe N=49, quindi tale istruzione diverrebbe: FI DSB 10,205.

Nel blocco AALBE sono contenuti gli $A_{\alpha\beta}$, mentre nel blocco TETA sono contenuti i ϑ_i (i = 0, 1, 2, 3, 4) che figurano nelle formule trovate. Essi vengono memorizzati nell'ordine: A_{10}, A_{20}, A_{21}, A_{30}, A_{31}, A_{32}, A_{40}, A_{41}, A_{42}, A_{43}, ϑ_0, ϑ_1, ϑ_2, ϑ_3, ϑ_4. Questi dati devono essere introdotti da nastro, mentre H, XO, N vanno introdotti da macchina da scrivere. Tutti i dati, sia d'ingresso che di uscita, sono in floating.

In uscita si ha: x, $\varphi(x)$. Per far sì che valori corrispondenti di x e $\varphi(x)$ siano scritti sulla stessa riga occorre porre un Tab. a 55.

Esempi numerici

Il metodo è stato provato su due equazioni integrali di cui si conosce la soluzione analitica.

La prima equazione è:

$$\varphi(x) = e^{-x^2}(1+x^2) + x - 1 + \int_0^x x^2 e^{-xs} \varphi(s)\, ds \tag{a}$$

Essa è a nucleo lineare (la funzione sotto il segno d'integrale dipende linearmente da $\varphi(s)$) e la sua soluzione analitica è: $\varphi(x) = x$.

Come altro esempio si è considerata l'equazione a nucleo non lineare:

$$\text{(b)} \qquad \varphi(x) = 1 - x + \int_0^x \left[x\, e^{s(x-2s)} + e^{-2s^2} \right] \varphi^2(s)\, ds$$

la cui soluzione analitica è $\varphi(x) = e^{x^2}$.

Riportiamo nelle tabelle allegate i risultati relativi ai due casi per il confronto (v. Tab. I-II-III).

Esame dei risultati ottenuti

Il programma, come già accennato, è stato provato sia per l'equazione (a) che per quella (b) introducendo successivamente i coefficienti di Runge, di Kutta e di Gill.

Il calcolo è stato eseguito con otto cifre significative, usando un passo H = 0.1.

Come si vede dai risultati riportati nelle tabelle I e II, rispettivamente per l'equazione (a) e (b), lo errore che si commette usando i vari coefficienti è pressocchè lo stesso. Ciò risulta evidente specialmente nel caso dei coefficienti di Runge e di Gill.

Il programma, inoltre, è stato provato, per l'equazione (a), usando 16 cifre significative. Come si vede dalla Tab. III, in pratica è sufficiente eseguire il calcolo con 8 cifre significative.

Desidero ringraziare vivamente il dott. A. Turrin per i preziosi consigli e l'incoraggiamento datimi durante la stesura di questo lavoro e la dott.ssa S. Josi per l'aiuto prestatomi.

Bibliografia

- Symposium on the numerical treatment of ordinary differential equations, integral and integro-differential equations (Rome, 20-24 September 1960) Pag. 362 - 363.
- Comptes Rendus des Séances de l'Académie des Sciences 1960 Vol. 250 - Pag. 3269 - 3270.
- Ralston - Wilf - Mathematical Methods for Digital Computers - Pag. 110 e seg.

TABELLA I

Equazione (a) - Soluzione analitica $\varphi(x) = x$ - H = 0.1

x	$\varphi(x)$ (coeff.Runge)	$\varphi(x)$ (coeff.Kutta)	$\varphi(x)$ (coeff.Gill)
0.1	0.10000005	0.10000005	0.10000005
0.2	0.20000004	0.20000004	0.20000005
0.3	0.30000003	0.30000002	0.30000002
0.4	0.40000002	0.39999997	0.40000001
0.5	0.49999996	0.49999988	0.49999997
0.6	0.59999998	0.59999992	0.59999999
0.7	0.70000000	0.69999988	0.69999999
0.8	0.79999991	0.79999981	0.79999991
0.9	0.89999987	0.89999976	0.89999988
1.0	0.99999986	0.99999969	0.99999985
1.1	1.0999993	1.0999992	1.0999994
1.2	1.1999989	1.1999987	1.1999990
1.3	1.2999986	1.2999983	1.2999985
1.4	1.3999978	1.3999977	1.3999979
1.5	1.4999971	1.4999972	1.4999971
1.6	1.5999968	1.5999965	1.5999966
1.7	1.6999956	1.6999963	1.6999954
1.8	1.7999957	1.7999954	1.7999955
1.9	1.8999949	1.8999952	1.8999950
2.0	1.9999942	1.9999945	1.9999940

TABELLA II

Equazione (b) - Soluzione analitica: $\varphi(x) = e^{x^2}$ - H = 0.1

x	e^{x^2}	$\varphi(x)$ (coeff.Runge)	$\varphi(x)$ (coeff.Kutta)	$\varphi(x)$ (coeff.Gill)
0.1	1.0100502	1.0100501	1.0100500	1.0100501
0.2	1.0408108	1.0408106	1.0408105	1.0408106
0.3	1.0941743	1.0941739	1.0941739	1.0941739
0.4	1.1735109	1.1735100	1.1735102	1.1735100
0.5	1.2840254	1.2840236	1.2840245	1.2840236
0.6	1.4333294	1.4333260	1.4333278	1.4333259
0.7	1.6323162	1.6323098	1.6323138	1.6323100
0.8	1.8964809	1.8964696	1.8964768	1.8964698
0.9	2.2479080	2.2478888	2.2479010	2.2478889
1.0	2.7182818	2.7182494	2.7182702	2.7182493

TABELLA III

Equazione (a) - Soluzione analitica $\varphi(x) = x$ - $H = 0,1$ - Calcolo eseguito con 16 cifre significative.

x	$\varphi(x)$ (coeff.Runge)	$\varphi(x)$ (coeff.Kutta)
0.1000000000000000	0.1000000000121700	0.0999999954556308
0.2000000000000000	0.2000000002998687	0.1999999967029444
0.3000000000000000	0.3000000018968862	0.2999999906131140
0.4000000000000000	0.4000000064953606	0.3999999825125058
0.5000000000000000	0.5000000152519867	0.4999999747221675
0.6000000000000000	0.6000000271754068	0.5999999686788391
0.7000000000000000	0.7000000383436591	0.6999999634526053
0.8000000000000000	0.8000000426723655	0.7999999558083483
0.9000000000000000	0.9000000337697062	0.8999999416511082
1.1000000000000000	1.1000000006670090	0.9999999177547460


START
Leggi: H, X0, N
Aαβ, θβ
Ψ0(x0) = f(x0)
p = 0
p1 = 0
xp = x0 + pH
α = 1
x = xp + θα H
Fp(x) = f(x)
β = 0
p ≥ 1
sì
no
Ψ0(xp) = Ψ4(xp-1)
j = 0
j1 = 0
xj = x0 + jH
Fp(x) = Fp(x) + HBβG[x, xj + θβH, Ψβ(xj)]
β = β + 1
β ≤ 3
sì
no
β = 0
j = j + 1
j1 = j1 + 1
j ≤ p-1
sì
no
S = 0
S = S + AαβG[x, xp + θβH, Ψβ(xj)]
β = β + 1
β > α-1
no
sì
Ψα(xp) = HS + Fp(x)
α = α + 1
α > 4
no
sì
stampa:
x, Ψ4(xp)
p = p + 1
p > N
no
sì
p1 = p1 + 1

```
*EQUAZIONE INTEGRALE DI VOLTERRA
DUE     DC    10,5120000000
BG      DS    5
LETT    DS    10
        DC    1,@
H       DS    10
XO      DS    10
N       DS    10
ALFA    DC    5,1
A       DS    5
BETA    DC    5,0
B       DS    5
AALBE   DSB   10,10
INAAB   DSA   AALBE
TETA    DSB   10,5
INTETA  DSA   TETA
P       DC    10,0000000000
P1      DC    5,0
XP      DS    10
X       DC    11,0000000000@
XQ      DS    10
XA      DS    10
UNO     DC    10,5110000000
J1      DC    5,0
J       DC    10,0000000000
D       DS    5
Q       DS    10
XJ      DS    10
C       DS    5
XBJ     DS    10
L       DS    10
S       DS    10
I       DS    5
B1      DS    5
XBP     DS    10
XB      DS    10
T       DS    10
ZERO    DC    10,0000000000
FI      DSB   10,85
INFI    DSA   FI
E       DS    10
FP      DS    10
R       DC    1,@
M       DS    5
Z       DS    10
G       DS    10
START   RNTY  H-9
        RCTY
        RNTY  XO-9
```

```
        RCTY
        RNTY N-9
        RCTY
MAMMA   TF   A,ALFA
        SM   A,1
        TF   C,A
        M    A,ALFA
        SF   95
        TF   A,99
        M    A,5
        SF   95
        TF   A,99
SILVIO  MM   BETA,10
        SF   95
        TF   B,99
        A    B,A
        A    B,INAAB
        RNPT LETT-9
        TF   *+18,B
        TF   ,LETT
        AM   BETA,1
        C    BETA,C
        BNH  SILVIO
        AM   ALFA,1
        CM   ALFA,4
        BH   LETCOM
        TFM  BETA,0
        B    MAMMA
LETCOM  TFM  ALFA,0
NVAL    MM   ALFA,10
        SF   95
        TF   A,99
        A    A,INTETA
        RNPT LETT-9
        TF   *+18,A
        TF   ,LETT
        AM   ALFA,1
        CM   ALFA,4
        BNH  NVAL
        TF   XA,XO
        BT   CALF,XO
        TF   FI,L
NP      FM   P,H

        TF   XP,99
        FA   XP,XO

        TF   Q,P
        FS   Q,UNO
```

```
        TFM   ALFA,1
NALFA   TFM   BETA,0
        TFM   J1,0
        TF    J,ZERO
        MM    ALFA,10
        SF    95
        TF    B,99
        A     B,INTETA
        TF    *+35,B
        FM    ,H

        TF    X-1,99
        FA    X-1,XP

        TF    XA,X-1
        BT    CALF,X-1
        TF    FP,L
        C     P,UNO
        BN    PPUNTO
NJ      MM    J1,40
        SF    95
        TF    D,99
        FM    J,H

        TF    XJ,99
        FA    XJ,XO

CBETA   MM    BETA,10
        SF    95
        TF    B,99
        TF    C,B
        AM    C,60
        A     C,INAAB
        TF    A,B
        A     A,INTETA
        TF    *+35,A
        FM    ,H

        TF    XBJ,99
        FA    XBJ,XJ

        TF    XB,XBJ
        BT    CALG,XB
        TF    *+35,C
        FM    ,G

        TF    L,99
        FM    L,H
```

```
       TF     L,99
       FA     FP,L

       AM     BETA,1
       CM     BETA,3
       BNH    CBETA
       TFM    BETA,0
       FA     J,UNO

       AM     J1,1
       C      J,Q
       BNH    NJ
PPUNTO TF     S,ZERO
       TF     I,ALFA
       SM     I,1
       M      I,ALFA
       SF     95
       TF     M,99
       MM     M,5
       SF     95
       TF     M,99
CONSUM M      BETA,10
       SF     95
       TF     B,99
       MM     P1,40
       SF     95
       TF     D,99
       TF     B1,B
       A      B1,INTETA
       TF     *+35,B1
       FM     ,H

       TF     XBP,99
       FA     XBP,XP

       TF     XB,XBP
       BT     CALG,XBP
       A      B,M
       A      B,INAAB
       TF     *+35,B
       FM     ,G

       TF     T,99
       FA     S,T

       AM     BETA,1
       C      BETA,I
       BNH    CONSUM
       MM     ALFA,10
       SF     95
```

```
        TF    A,99
        A     A,D
        A     A,INFI
        FM    H,S

        TF    E,99
        FA    E,FP

        TF    *+18,A
        TF    ,E
        AM    ALFA,1
        CM    ALFA,4
        BNH   NALFA
        AM    A,1
        TF    *+18,A
        TD    ,R
        WNTY  X-10
        TBTY
        SM    A,10
        TF    *+18,A
        WNTY
        RCTY
        FA    P,UNO

        C     P,N
        BH    RIPRI
        AM    P1,1
        B     NP
        NOP
CALF    TF    L,UNO
        FS    L,XA

        BB
CALG    FM    XB,X-1

        TF    T,99
        FEX   L,T

        FM    L,X-1

        TF    L,99
        FA    L,UNO

        TF    XQ,XB
        FM    XQ,XB

        TF    XQ,99
        FM    XQ,DUE
```

```
         TF    XQ,99
         FEX   T,XQ

         FD    L,T

         TF    L,99
         TF    BG,B
         A     BG,D
         A     BG,INFI
         TF    *+23,BG
         TF    T,
         TF    *+35,BG
         FM    ,T

         TF    T,99
         FM    T,L

         TF    G,99
         BB
RIPRI    TFM   ALFA,1
         TFM   BETA,0
         TF    P,ZERO
         TFM   P1,0
         B     START
         DEND  START
```